CADSys Tutorial





Introduction





CADSys is a powerful and flexible 2D vectorial graphics engine. It hasn't a lot of methods and properties but it can be used for many projects that need graphics. It uses the OOP technology to make a system that can be expanded with a little effort. In this tutorial you will learn how to use the library, in particular how to use the TViewportPanel. This visible component gives you a starting point on which to build your own application without bothering you about the view management (like zooming and panning).


The following sections will give you the necessary knowledge to use the library (I hope so). If you have any suggestion please send me an e-mail at VPIERO@IVG.IT





Section 1 - How to make a simple viewer


Section 2 - Adding printing and exporting functionality


Section 3 - How interacting with your drawings


Section 4 - Insert a wallpaper in your drawings








Section 1 - How to make a simple viewer





A viewer is a kind of tool that allow you to see a drawing on the screen. Normally you have a file of drawings, in a format similar to DXF, and you might want to show it in your programs. In this tutorial I will not show you how to read a particular format (in a future release a reader/writer of DXF files will be present), but how to show a drawing. Therefore I consider that you have the various primitives and you migtht want to draw them.





The first step is to create a new project in Delphi. When you have the form you must add the TViewportPanel and TCADCmp, dragging it from the CADCmp palete on the form.


Now set the Align property of TViewportPanel1 to alClient and the CADCmp2D to CADCmp1. This will link the CAD component to the viewport. After that, you can personalize the viewport with the various property. For example the MaintainAspect set to true insure that your view will be always with the correct aspect ratio. You can also control what buttons will be present on the toolbar, using the ShowToolbar property. If this property is set to [], then no toolbar will be drawn. Don't forget to add the following libraries to the use clause:





	GrLib2D, Prim2D, Obj2D, Block2D





We will use some functions from them.Your form will be like this:





�embed PaintShopPro Object1 ���





Note that I have added some other controls, two buttons and one panel. The viewer is done. The only step that remain to do is to draw the drawing. This can be accomplished with a function called for example from a button or a menu. For simplicity I draw a sample drawing in the OnCreate method of the form:





procedure TForm1.FormCreate(Sender: TObject);


var


  Tmp: TObject2D;


begin


  Viewportpanel1.Extensions := Rect2D(-1, -1, 101, 101);


  with CADCmp1 do


   begin


     AddObject(-1, TFrame2D.Create(0, Point2D(0, 0), Point2D(100, 100)));


     Tmp := TText2D.Create(0, Rect2D(0, 0, 100, 99), 20, 'Hello word');


     TText2D(Tmp).DrawBox := False;


     TText2D(Tmp).ClippingFlags := DT_CENTER;


     AddObject(-1, Tmp);


   end;


end;





The first statament sets the Extensions of the drawing area. This property was added only in TViewportPanel, and is used to set the two scrollbar used for paning. You can set any value you want, no check will be made on the extensions. So you can draw a primitive outside that area. After that two object will be added to CADCmp1. When you want to draw somethig you must draw on a CADCmp and not directly on a viewport. This made your object persistent.


The first object is frame from (0, 0) to (100, 100). Note the calling metod. AddObject require two parameters, the first is the object ID and the second the Object itself. If the ID is less than zero, the library will select an unique ID.


The second object is a Text. Because we want to set some properties of the text before adding it to the CADCmp, we use a temporary object (that we can save with a try-finally block). After setting the properties we add the component. Note that we don't free the temporary object. This is an assignament of the CADCmp.





That's all, if you run the program you will see the drawing. Use the program a little so you take confidence with the library.





If you have runned the project, you have seen that the initial view is somewhat strange. In fact the default view for TViewportPanel is the window from (-100, -100) to (100, 100). Not to good. Almost the times, we want to start with some other view, for example the extensions view, or the all view, or something else. How to make this possible ?


It's very simple. Add the following statament at the end of the OnCreate method:





	ViewportPanel1.Viewport.ZoomWindow(Rect2D(0, 0, 100, 100));





Or:





	ViewportPanel1.ZoomAllClick(Self);





Note that if you have set MaintainAspect to True, your result will not exactly what you want. The reason is that the view will be resized to maintain the aspect ratio.





Section 1.1





We have seen how to view a drawing on the screen. Not too diffucult, Do you belive ? Obvioulsy, this project is too simple to be usefull. So let add some functionality to it.


Firstly we need a way to save and load the drawing in an internal format. This let us to store our drawing in a more conpact form respect to a DXF file for example.


Again the library will do the hard work. To begin add a file menu with load and save function,


and an Open and Save dialog boxes.


The in the two menu event add the following code:





procedure TForm1.Load1Click(Sender: TObject);


var


  TmpStream: TFileStream;


begin


  if OpenDialog1.Execute then


   begin


     TmpStream := TFileStream.Create(OpenDialog1.FileName, fmOpenRead);


     try


       CADCmp1.LoadFromStream(TmpStream);


       ViewportPanel1.ZoomAllClick(Self);


     finally


       TmpStream.Free;


     end;


   end;


end;





procedure TForm1.Save1Click(Sender: TObject);


var


  TmpStream: TFileStream;


begin


  if SaveDialog1.Execute then


   begin


     TmpStream := TFileStream.Create(SaveDialog1.FileName, fmOpenWrite or fmCreate);


     try


      CADCmp1.SaveToStream(TmpStream);


     finally


       TmpStream.Free;


     end;


   end;


end;





It may seem strang to use a stream instead of a file, but this can be usefull if you don't have a file, but a memory region. In a usefull program there will be also a way to create/modify the drawing or to import it. In our example is usefullness to save a drawing that will be created all the time by the program.However if you comment-out the OnCreate method you will see that the method will work. Note that the CADCmp will delete the current drawing before to load a new one.





Another usefull function of our example will be the capability to do the same task of the toolbar from a popup menu. So add a popup component to the form and set the PoupUpMenu property of TViewportPanel to it. Then add the following command to the popup:





�embed PaintShopPro Object1 ���





Now add the following code to the menu events:





procedure TForm1.ZoomIn1Click(Sender: TObject);


begin


  ViewportPanel1.ZoomInClick(Sender);


end;





procedure TForm1.ZoomOut1Click(Sender: TObject);


begin


  ViewportPanel1.ZoomOutClick(Sender);


end;





procedure TForm1.ZoomAll1Click(Sender: TObject);


begin


  ViewportPanel1.ZoomAllClick(Sender);


end;





procedure TForm1.ZoomExtension1Click(Sender: TObject);


begin


  ViewportPanel1.ZoomToExtensionClick(Sender);


end;





There is a problem however. If your drawing is complex, it can take a time to redraw it. If you take a look to the toolbar, when this is the case, the buttons of the zooming are disabled. Even if zooming during a repaint is not a problem (it will stop the current redrawing if the UseProcessMessages of the Viewport is set to true) retain our popup menu enabled during a repaint can be unaesthetic. So let us to use another aid of the TViewportPanel. Double click on the event OnBeginRedraw and OnEndRedraw of TViewportPanel and add the following code:





procedure TForm1.ViewportPanel1BeginRedraw(Sender: TObject);


begin


  PopUpMenu1.AutoPopUp := False;


end;





procedure TForm1.ViewportPanel1EndRedraw(Sender: TObject);


begin


  PopUpMenu1.AutoPopUp := True;


end;





Now our popup will has the same semantic of the toolbar.


A final improvement to our example, is to add a new menu that permit to change the actual setting of MaintainAspec and show grid. To do that add a new menu, and two item. In the event handler of that two item type:





procedure TForm1.Maintainaspectratio1Click(Sender: TObject);


begin


  with ViewportPanel1 do


   begin


     MaintainAspect := not MaintainaspectRatio1.Checked;


     MaintainaspectRatio1.Checked := MaintainAspect;


     Viewport.Repaint;


   end;


end;





procedure TForm1.Showgrid1Click(Sender: TObject);


begin


  with ViewportPanel1.Viewport do


   begin


     if Showgrid1.Checked then


      GridStep := Point2D(0, 0)


     else


      GridStep := Point2D(10, 10);


     Repaint;


   end;


  Showgrid1.Checked := not Showgrid1.Checked;


end;





Furthermore in the OnCreate method of the form:





  MaintainaspectRatio1.Checked := ViewportPanel1.MaintainAspect;


  if ViewportPanel1.Viewport.GridStep.X > 0.0 then


   Showgrid1.Checked := True;





We have finished. Our example is a complete viewer. You can also add a function to change the layer settings using the Layers property of CADCmp. However we don't see that here. See the example1 project for details.





 NOTE: To simulate the functions of the toolbar in a popup menu, we have create the events handlers for the menu's items and in these handlers we have called the event's handlers in the ViewportPanel components. There is a more usefull way to do that. In the OnCreate metod of the viewer add the following code:





  with ViewportPanel1 do


   begin


     ZoomIn1.OnClick := ZoomInClick;


     ZoomOut1.OnClick := ZoomOutClick;


     ZoomAll1.OnClick := ZoomAllClick;


     ZoomExtension1.OnClick := ZoomToExtensionClick;


   end;





Section 2 - Adding printing and exporting functionality





Now we have a simple but complete viewer, and we have obtained it without a lot of effort ! Now we must add some important functionality, that are necessary for a serious and usefull viewer. When we have our drawing on the screen and have zoommed or modified it, we must be able to print it or to export it as a bitmap or a metafile. In this section we will see how to do that. Let's start with the printing capabilities.


The Viewport of CADSys library has a metod that can be used to implement printing and exporting. It's name is CopyOnCanvas. It's goal is to allow you to copy the contents of the viewport on an external canvas. So the only thing to do is to pass the canvas of the Printer object to CopyOnCanvas to obtain an hard copy of the drawing.


So, Add a new items in the file menu and call it Print.The add a submenu of Print with the options: In its event handler type the following code:





procedure TForm1.All1Click(Sender: TObject);


begin


  Printer.Orientation := poLandScape;


  Printer.BeginDoc;


  try


    with ViewportPanel1.Viewport do


     begin


       ViewportPanel1.BackGroundColor := clWhite;


       CopyOnCanvas(Printer.Canvas, cmExtensions, 1.0, 1.0);


       ViewportPanel1.BackGroundColor := clSilver;


     end;


  finally


    Printer.EndDoc;


  end;


end;





procedure TForm1.Fitcurrent1Click(Sender: TObject);


begin


  Printer.Orientation := poLandScape;


  Printer.BeginDoc;


  try


    with ViewportPanel1.Viewport do


     begin


       ViewportPanel1.BackGroundColor := clWhite;


       CopyOnCanvas(Printer.Canvas, cmFit, 1.0, 1.0);


       ViewportPanel1.BackGroundColor := clSilver;


     end;


  finally


    Printer.EndDoc;


  end;


end;





procedure TForm1.Scale1Click(Sender: TObject);


begin


  Printer.Orientation := poLandScape;


  Printer.BeginDoc;


  try


    with ViewportPanel1.Viewport do


     begin


       ViewportPanel1.BackGroundColor := clWhite;


       CopyOnCanvas(Printer.Canvas, cmScale, 1.0, 1.0);


       ViewportPanel1.BackGroundColor := clSilver;


     end;


  finally


    Printer.EndDoc;


  end;


end;





Note that I have changed the background color from clSilver (the default) to clWhite. The same method can be used to export the current drawing to a bitmap or a Windows Metafile.


For example the following code is used to do that:





To Export to a bitmap





procedure TForm1.All2Click(Sender: TObject);


var


  TmpBmp: TBitmap;


begin


  TmpBmp := TBitmap.Create;


  try


    TmpBmp.Width := 200;


    TmpBmp.Height := 200;


    with ViewportPanel1.Viewport do


     begin


       ViewportPanel1.BackGroundColor := clWhite;


       CopyOnCanvas(TmpBmp.Canvas, cmExtensions, 0.0, 0.0);


       ViewportPanel1.BackGroundColor := clSilver;


     end;


    Clipboard.Assign(TmpBmp);


  finally


    TmpBmp.Free;


  end;


end;





procedure TForm1.Current1Click(Sender: TObject);


var


  TmpBmp: TBitmap;


begin


  TmpBmp := TBitmap.Create;


  try


    TmpBmp.Width := 200;


    TmpBmp.Height := 200;


    with ViewportPanel1.Viewport do


     begin


       ViewportPanel1.BackGroundColor := clWhite;


       CopyOnCanvas(TmpBmp.Canvas, cmFit, 0.0, 0.0);


       ViewportPanel1.BackGroundColor := clSilver;


     end;


     Clipboard.Assign(TmpBmp);


  finally


    TmpBmp.Free;


  end;


end;





In this code I have set the bitmap dimension to 200x200. Obviously you can set it to whatever value you want. Only remember that a larger bitmap require a large ammount of memory.





To Export to a metafile





procedure TForm1.All3Click(Sender: TObject);


var


  TmpMeta: TMetafile;


  TmpMetaCnv: TMetafileCanvas;


  RefDC: HDC;


begin


  TmpMeta := TMetafile.Create;


  RefDC := GetDC(0);


  try


    TmpMeta.Width := GetDeviceCaps(RefDC, HORZRES);


    TmpMeta.Height := GetDeviceCaps(RefDC, VERTRES);


    TmpMetaCnv := TMetafileCanvas.Create(TmpMeta, 0);


    try


     with ViewportPanel1.Viewport do


      begin


        ViewportPanel1.BackGroundColor := clWhite;


        CopyOnCanvas(TmpMetaCnv, cmExtensions, 1.0, 1.0);


        ViewportPanel1.BackGroundColor := clSilver;


      end;


    finally


     TmpMetaCnv.Free;


    end;


    Clipboard.Assign(TmpMeta);


  finally


    TmpMeta.Free;


    ReleaseDC(0, RefDC);


  end;


end;





procedure TForm1.Current2Click(Sender: TObject);


var


  TmpMeta: TMetafile;


  TmpMetaCnv: TMetafileCanvas;


  RefDC: HDC;


begin


  TmpMeta := TMetafile.Create;


  RefDC := GetDC(0);


  try


    TmpMeta.Width := GetDeviceCaps(RefDC, HORZRES);


    TmpMeta.Height := GetDeviceCaps(RefDC, VERTRES);


    TmpMetaCnv := TMetafileCanvas.Create(TmpMeta, RefDC);


    try


     with ViewportPanel1.Viewport do


      begin


        ViewportPanel1.BackGroundColor := clWhite;


        CopyOnCanvas(TmpMetaCnv, cmFit, 1.0, 1.0);


        ViewportPanel1.BackGroundColor := clSilver;


      end;


    finally


     TmpMetaCnv.Free;


    end;


    Clipboard.Assign(TmpMeta);


  finally


    TmpMeta.Free;


    ReleaseDC(0, RefDC);


  end;


end;





This code is only an example, I haven't yet understood how metafile works. I only see that if you create a TMetafile with the reference DC (the second parameter of Create method) equal to zero, the screen is used as DC. So you must set the metafile dimension to the dimension of the screen or the result will not be what you want. So I open the DC of the screen myself, assign the correct dimensions and then pass it to the Metafile Canvas. This work, but I don't know exactly why.





Section 3 - How interacting with your drawing





Now make a step forward. We have seen how to draw a drawing and how to export it. A large kind of applications need also to interact with the drawing, for example to move some objects or to add a quote. In this section we will see how this can be accomplished with CADSys library. This section will require more code respect the previous two, and it may be a little difficult to understand. However the library will do the more difficult task.


The first task we will see is how to move an object in the drawing. The first step to move an objec is to obtain an object to move. TViewport has a method, PickObject, that return the object at some point in the world. Using this method we can know when to start the moving and on what object do the moving.


First add a field in the form that maintain the last object picked:





  ....


  private


    { Private declarations }


    FLastPicked: TObject2D;


    FDragPoint: TPoint2D;


  ....





and initialize it in the OnCreate event to nil. The PickObject of TViewport have tree parameters:





Pt: TPoint2D		That contain the point in world coordinates at which we want to test if 


			an object exists.


Aperture: Integer	The size of the pick window.


NPoint: Integer		The point picked. This can be:


			 -2    if the mouse was clicked in the primitive (for closed polygon)	


			 -1    if the mouse was clicked on the primitive


			 > -1 if the mouse was clicked on a control point (Points[]) of the 


			        primitive.





The method return a TObject2D if an object was found or nil if no object was found. The point passed to PickObject can be obtained from the TMouseDown2D event of TViewportPanel. In this method the WX, WY parameters contain the point in world coordinate at which the mouse's button was clicked. In that event type the following code:





procedure TForm1.ViewportPanel1MouseDown2D(Sender: TObject;


  Button: TMouseButton; Shift: TShiftState; WX, WY: Single; X, Y: Integer);


var


  TmpInt: Integer;


begin


  if FLastPicked = nil then


   begin


     FLastPicked := ViewportPanel1.Viewport.PickObject(Point2D(WX, WY), 4,


                                                       TmpInt);


     FDragPoint := Point2D(WX, WY);


   end


  else


   begin


     FLastPicked := nil;


     ViewportPanel1.Viewport.Repaint;


   end;


end;





First we see if we are already moving an object. If so we repaint the drawing to see the object in the new position and set FLastPicked to nil. Otherwise we get the object at WX, WY (if any) and set the FDrawPoint to WX, WY. We see below the scope of FDragPoint.





Now we have the object picked and we want to move it. TObject2D class has a method, MoveTo, that can be used to move the object. It accepts two parameters, the first is the destination point of the move, the second the starting point of the move action. The two point must be in world coordinates. The first point can be retrived in the OnMouseMove2D the second is FDragPoint. The two point give the step by which the base point of the object will be moved. So we need to reset the starting point to the current point after we have moved the object. We must do another task. It's necessary to show the object when we are moving it. This is accomplished by the DrawObjectWithRubber of TViewport. This method draw the object desired only on the canvas on the screen using the function XOR. So if we draw an object two times in the same place with that function, the object will disappear after the second call. This is the heart of our moving function. When we move the mouse we first draw the object, then move it, and then redraw it in the new position. The first call show the object in it's starting position and it will be visible for all the time of the moving, the subsquent calls will show the object with a move fashion. The code to type in OnMouseMove2D is:





procedure TForm1.ViewportPanel1MouseMove2D(Sender: TObject;


  Shift: TShiftState; WX, WY: Single; X, Y: Integer);


begin


  if FLastPicked <> nil then


   begin


     ViewportPanel1.Viewport.DrawObjectWithRubber(FLastPicked, clRed);


     FLastPicked.MoveTo(Point2D(WX, WY), FDragPoint);


     ViewportPanel1.Viewport.DrawObjectWithRubber(FLastPicked, clRed);


     FDragPoint := Point2D(WX, WY);


   end;


end;





That's all ! If you run the viewer you will see that this code work, with one exception. If you move the text, the Rubber method will not work. The cause is that DrawText API function doesn't use the XOR method to draw a text. If you want to resolve this drawback, you can use another method. You must refresh (not repaint) the viewport before draw the moved object on it, and you must draw the object only on the canvas (setting the WhereToDraw property of TViewport to [wdCanvas]). To draw the object you must use the DrawObject method of TViewport. The new code is:





procedure TForm1.ViewportPanel1MouseMove2D(Sender: TObject;


  Shift: TShiftState; WX, WY: Single; X, Y: Integer);


begin


  if FLastPicked <> nil then


   with ViewportPanel1.Viewport do


    begin


      Refresh;


      WhereToDraw := [wdCanvas];


      FLastPicked.MoveTo(Point2D(WX, WY), FDragPoint);


      DrawObject(FLastPicked);


      WhereToDraw := [wdOffScreen];


      FDragPoint := Point2D(WX, WY);


   end;


end;





The drawback of this method is a more visible flickering of the object moved. However move the object is not to difficult !





Another usefull function is the ability to quote a primitive. For simplicity we will see only the lines quoting, but is a simple exercise expand that function on others primitives.


A quote is a line that go from the start point of a line to the end point. On that line a number is present, which indicates the lenght of the line. What we need is to draw these objects after we have picked an object. The steps to obtain that are really similar to the steps of moving a primitive. So we will use the same events. 


First of all make a function that, given a line, will draw the quote.





procedure TForm1.DrawQuote(Line: TLine2D; Dist: TRealType);


var


  TmpCont: TContainer2D;


  TmpText: TText2D;


  Pt1, Pt2: TPoint2D; { Quote line points. }


  Rect1: TRect2D;


  Ang, Len, DistX, DistY: TRealType;


begin


  if Line.Tag <> 0 then Exit;


  Pt1 := Line.Points[0];


  Pt2 := Line.Points[1];


  if Pt2.Y - Pt1.Y <> 0.0 then


   begin


     Ang := ArcTan(-(Pt2.X - Pt1.X) / (Pt2.Y - Pt1.Y));


     Len := Sqrt(SumOfSquares([Pt2.X - Pt1.X, Pt2.Y - Pt1.Y]));


     DistX := Dist / cos(Ang);


     DistY := Dist / sin(Ang);


     Rect1.Left := Pt1.X - DistX;


     Rect1.Bottom := Pt1.Y - DistY;


     Rect1.Right := Pt2.X - DistX;


     Rect1.Top := Pt2.Y - DistY;


   end


  else


   begin


     Ang := 0;


     Len := Abs(Pt2.X - Pt1.X);


     Rect1.Left := Pt1.X;


     Rect1.Bottom := Pt1.Y + Dist;


     Rect1.Right := Pt2.X;


     Rect1.Top := Pt2.Y + Dist;


   end;


  CADCmp1.CurrentLayer := 240;


  TmpText := TText2D.Create(0, Rect1, 5, Format('%f', [Len]));


  TmpText.DrawBox := False;


  TmpText.ClippingFlags := DT_CENTER or DT_VCENTER or DT_SINGLELINE


                           or DT_NOCLIP;


  TmpCont := TContainer2D.Create(0,


              [TLine2D.Create(0, Rect1.FirstEdge, Rect1.SecondEdge),


               TLine2D.Create(1, Pt1, Rect1.FirstEdge),


               TLine2D.Create(2, Pt2, Rect1.SecondEdge),


               TmpText]);


  TmpCont.Enabled := False;


  TmpCont.ModelTransform := Line.ModelTransform;


  CADCmp1.AddObject(-1, TmpCont);


  Line.Tag := LongInt(TmpCont);


  Line.OnChange := MoveQuote;


  with ViewportPanel1.Viewport do


   begin


     DrawObject(TmpCont);


     Refresh;


   end;


end;





This is only an example, the algoritm used here is not the best and it's not alwais correct (i don't check in which quadrant falls the Angle before using ArcTan). However it's a starting point. The interesting part of the algoritm is the use of TContainer2D and the OnChange event of TObject2D. TContainer2D is a primitive used to contain other 2D objects. It's similar to TSourceBlock2D used for blocks definitions, but it hasn't a reference count. Becouse we need to deal with the various elements of a quote (lines and text) altogether, we need to pack these elements in a container. Also we set the Enabled property of the container to False, so we cannot move the quote (the quote is connected to the object quoted). However we can move the quoted line, so we need a mechanism to move the quote with the line. First we set the ModelTransform of the container to the ModelTransform of the line, so the two object will be sincronized (remember that Points property of TPrimitive2D is in object coordinates).


The we use two powerfull property of TObject2D. First we store in the Tag property the reference to the quote's container. When we pick a line we know also the associated quote. The we set the OnChange event of the line to the function MoveQuote:





procedure TForm1.MoveQuote(Sender: TObject);


var


  Cont2D: TContainer2D;


  TmpLine: TLine2D;


begin


  if (Sender is TLine2D) and (TLine2D(Sender).Tag <> 0) then


   begin


     Cont2D := TContainer2D(TLine2D(Sender).Tag);


     Cont2D.ModelTransform := TLine2D(Sender).ModelTransform;


   end;


end;





The OnChange event is called whenever we change the object (move it, transform it, and so on). In the event handler we first check if the Sender is a line and the if the Tag is assigned. If this is the case, we know that the line as a quote, so we apply the same transform of the line to the container. This will reset the quote properties. 





Note: If we want to have more than one quote on a line, we must declare a new class that contains all the quotes, and set the Tag property to the reference of that class.





The last thig to to is to add a menu command that start a quoting action, and pick up the object to be quoted:





procedure TForm1.ViewportPanel1MouseDown2D(Sender: TObject;


  Button: TMouseButton; Shift: TShiftState; WX, WY: Single; X, Y: Integer);


var


  TmpInt: Integer;


begin


  if FLastPicked = nil then


   begin


     FLastPicked := ViewportPanel1.Viewport.PickObject(Point2D(WX, WY), 4,


                                                       TmpInt);


     FDragPoint := Point2D(WX, WY);


     if FOnQuote and (FLastPicked is TLine2D) then


      begin


        DrawQuote(FLastPicked as TLine2D, 5);


        FLastPicked := nil;


        FOnQuote := False;


      end;


   end


  else


   begin


     FLastPicked := nil;


     ViewportPanel1.Viewport.Repaint;


   end;


end;





The code that we have done, is almost correct. However if you quote a scaled line, the quote attacked will be also scaled. This happen because we set the ModelTransform of the two object to be the same.


A better way to do our job (also more complex) is given below:





procedure AdjustQuote(Line: TLine2D; Dist: TRealType);


var


  TmpCont: TContainer2D;


  Pt1, Pt2: TPoint2D;


  Rect1: TRect2D;


  Ang, Len, DistX, DistY: TRealType;


begin


  if Line.Tag = 0 then Exit;


  TmpCont := TContainer2D(Line.Tag);


  Pt1 := TransformPoint2D(Line.Points[0], Line.ModelTransform);


  Pt2 := TransformPoint2D(Line.Points[1], Line.ModelTransform);


  if Pt2.Y - Pt1.Y <> 0.0 then


   begin


     Ang := ArcTan(-(Pt2.X - Pt1.X) / (Pt2.Y - Pt1.Y));


     Len := Sqrt(SumOfSquares([Pt2.X - Pt1.X, Pt2.Y - Pt1.Y]));


     DistX := Dist / cos(Ang);


     DistY := Dist / sin(Ang);


     Rect1.Left := Pt1.X - DistX;


     Rect1.Bottom := Pt1.Y - DistY;


     Rect1.Right := Pt2.X - DistX;


     Rect1.Top := Pt2.Y - DistY;


   end


  else


   begin


     Ang := 0;


     Len := Abs(Pt2.X - Pt1.X);


     Rect1.Left := Pt1.X;


     Rect1.Bottom := Pt1.Y + Dist;


     Rect1.Right := Pt2.X;


     Rect1.Top := Pt2.Y + Dist;


   end;


  with TmpCont do


   begin


     with TLine2D(Objects.First) do


      begin


        Points[0] := Rect1.FirstEdge;


        Points[1] := Rect1.SecondEdge;


      end;


     with TLine2D(Objects.Next) do


      begin


        Points[0] := Pt1;


        Points[1] := Rect1.FirstEdge;


      end;


     with TLine2D(Objects.Next) do


      begin


        Points[0] := Pt2;


        Points[1] := Rect1.SecondEdge;


      end;


     with TText2D(Objects.Next) do


      begin


        Points[0] := Rect1.FirstEdge;


        Points[1] := Rect1.SecondEdge;


        Text := Format('%f', [Len]);


      end;


     UpdateExtension;


   end;


end;





procedure TForm1.MoveQuote(Sender: TObject);


begin


  if (Sender is TLine2D) then


   AdjustQuote(TLine2D(Sender), 5);


end;





procedure TForm1.DrawQuote(Line: TLine2D; Dist: TRealType);


var


  TmpCont: TContainer2D;


  TmpText: TText2D;


begin


  if Line.Tag <> 0 then Exit;


  CADCmp1.CurrentLayer := 240;


  TmpText := TText2D.Create(0, Rect2D(0, 0, 0, 0), 5, '');


  TmpText.DrawBox := False;


  TmpText.ClippingFlags := DT_CENTER or DT_VCENTER or DT_SINGLELINE


                           or DT_NOCLIP;


  TmpCont := TContainer2D.Create(0,


              [TLine2D.Create(0, Point2D(0, 0), Point2D(0, 0)),


               TLine2D.Create(1, Point2D(0, 0), Point2D(0, 0)),


               TLine2D.Create(2, Point2D(0, 0), Point2D(0, 0)),


               TmpText]);


  TmpCont.Enabled := False;


  Line.Tag := LongInt(TmpCont);


  AdjustQuote(Line, Dist);


  Line.OnChange := MoveQuote;


  CADCmp1.AddObject(-1, TmpCont);


  with ViewportPanel1.Viewport do


   begin


     DrawObject(TmpCont);


     Refresh;


   end;


end;





Now we have a better solution. However it has yet problems.





Section 4 - Insert a wallpaper in your drawing





If you want to insert a bitmap in your drawing that will be resized when you are zooming, you must use a new type of primitive: TBitmap2D. Actually this primitive has a restriction: If you zoom very close, the bitmap will not be drawed. However at these zoom the bitmap can be very coarse. However to add a wallpaper you can use this code:





procedure TForm1.SetWallpaper1Click(Sender: TObject);


var


  TmpBmp: TBitmap;


  Tmp: TObject2D;


begin


  if OpenDialog2.Execute then


   begin


     TmpBmp := TBitmap.Create;


     try


       TmpBmp.LoadFromFile(OpenDialog2.FileName);


       with ViewportPanel1.Extensions do


        Tmp := TBitmap2D.Create(0, FirstEdge, SecondEdge, TmpBmp);


       with CADCmp1 do


        InsertObject(-1, ObjectsList.First.ID, Tmp);


       ViewportPanel1.Viewport.Repaint;


     finally


       TmpBmp.Free;


     end;


   end;


end;





Note that we insert the new object in front of the first object in CADCmp1. This ensure that the bitmap will be drawed as a background.





CONCLUSION





In this tutorial we have seen how to use the library. However you must read the help file to see all the functions present in it. If you don't want to type the code above, you can find the viewer in the subdirectory Examples/Example1 of the distributions. Good luck.








						Piero Valagussa


